
This rare user manual was turning to dust in my shed. It has been recreated here for archival purposes
and distributed under fair use due to its very limited effect on the market value of this SPEECH
SYSTEMS product. Due to the deterioration of the printed paper, it has been re-typeset. Some minor
edits have been made for clarity as some spots were difficult to follow and begged for some editing.
Also, due to changes in Daylight Savings adjustment dates, the chip no longer updates at the correct
times. In the 1980’s, Motorola marked the MC146818 "NOT RECOMMENDED FOR NEW DESIGNS" and
afterward, Motorola made the MC146818 chip obsolete. DALLAS/MAXIM do make a replacement for
this part under numbers DS12885, DS12887, DS1685, or DS1687. With a small adjustment, these parts
can be used in place of the MC146818. [2/28/2018 - PW]

Copyright Notice
This entire manual along with all the computer programs are supplied for the personal use of the purchaser. In addition,
the contents of this manual are copyrighted by SPEECH SYSTEMS and reproduction by any means is strictly prohibited.

Warranty
SPEECH SYSTEMS warranties the PRECISION TIME MODULE against defects in material and workmanship for a period of
Ninety Days from the date of purchase to the original purchaser. The obligation of SPEECH SYSTEMS is limited to the
repair or replacement of the product, free of all charges, which proves defective during this period. This warranty does
not cover damage due to accident, negligence, abuse ot tampering.

Disclaimer
SPEECH SYSTEMS makes no other warranties or guarantees, express, statutory or implied, of any kind whatsoever with
respect to the product purchased, and any other implied warranty of merchantability or fitness for a particular purpose is
hereby disclaimed by SPEECH SYSTEMS and excluded from any agreement made by SPEECH SYSTEMS. The product is sold
on an "as is" basis. SPEECH SYSTEMS will not be responsible for any damage of any kind not covered by the exclusive
remedies set forth in this limited warranty. SPEECH SYSTEMS will not be responsible for any special, indirect or
consequential damage caused by its products.

SPEECH SYSTEMS reserves the right to make changes to any products or specifications described in this manual without
notification.

Precision Time Module

Moving Tape to Disk
All programs supplied with the PRECISION TIME MODULE are in BASIC. Therefore, transferring these programs to disk is
very simple.

1. Insert the tape and press play on the tape player.

2. Type CLOAD followed by |ENTER| and the first program will load.

3. Now type SAVE "PROGRAM" and the program will be saved to disk.

4. Repeat until all programs are saved.

Disk Owners
The PRECISION TIME MODULE is a hardware device that normally fits into the slot reserved for the disk controller.
Therefore disk owners will require some type of expansion device to allow simultaneous connection of the PRECISION
TIME MODULE and a disk controller.

The PRECISION TIME MODULE will work in all expansion units that we know of. It will work perfectly well with SPEECH
SYSTEMS Y-CABLE product. The Y-CABLE is a cable that contains a connector that plugs into the Color Computer and at
the other end are two connectors just like the one that is inside the Color Computer. You insert your disk controller into
one connector and the PRECISION TIME MODULE into the other.

If the Radio Shack Multi-Pak is used, the PRECISION TIME MODULE may be placed into any slot. If you use the Multi-Pak
with the PRECISION TIME MODULE you do not need place the slot control switch on the front of the Multi-Pak in any
particular position. In fact, you probably have your disk controller in slot 4 and the switch in position 4 and you can leave it
that way. In other words, the PRECISION TIME MODULE is active at all times.


Owners who use a BT-1000 or the BT-2000 by BASIC TECHNOLOGY, will have to add an integrated
circuit on the PRECISION TIME MODULE board. In the lower left hand corner of the PRECISION TIME
MODULE is a place for a 7407 IC. You may insert this chip yourself, but we suggest you call SPEECH
SYSTEMS first since opening the PRECISION TIME MODULE will void the warranty.

Moving Tape to Disk

Getting Started
Insert the PRECISION TIME MODULE in the Color Computer directly or in an expansion unit if you are using a disk system.
Load and RUN the program SETTIME to set the current time and date. Then load and RUN the CLOCK program. A description
and explanation of these programs follows.

Program Descriptions
There are 3 BASIC programs supplied with your PRECISION TIME MODULE, they are: SETTIME, CLOCK, and NUCLEUS. The
source listings are enclosed as a convenience. It is our hope that with an examination of these programs, together with
this manual, you will have the necessary information to properly use the PRECISION TIME MODULE.

SETTIME
This program allows you to set the time and date on the module. When the PRECISION TIME MODULE leaves the factory,
it is preset and we hope that you should not have to set it again. However, from time to time you may find this necessary.
When the program is run, you are prompted to enter the time and date.

 The program assumes that the hour value is entered using the 24 hour format, which is a value from 0
to 23.

For example, 2:00 in the afternoon is 14 hundred hours and 2:00 in the morning is 2 hundred hours.

The PRECISION TIME MODULE is initialized as a 24 hour clock rather than a 12 hour clock with AM and PM indications. If
you wish to modify the program to initialize the PRECISION TIME MODULE as a 12 hour clock, we direct you to the data
sheet sheet excerpts for the MC146818 in this manual. It describes how to program the registers as a 12 hour clock.

The program will initialize the PRECISION TIME MODULE to use Daylight Savings Time. This means that on the last Sunday
in April the PRECISION TIME MODULE increments from 1:59:59 AM to 3:00:00 AM. In addition, on the last Sunday in
October when the time first reaches 1:59:59 AM it changes to 1:00:00 AM.

 Daylight Savings periods in the United States were changed in 2007, making the DST adjustment dates
in the MC146818 no longer correct.

These special updates do not occur when the DSE bit is set to a 0. Once again we direct you to the data sheet on the
MC146818 if you wish to modify the initialization steps that the SETTIME program uses.

Precision Time Module

4



100 '       SETTIME
110 '     BY RICH PARRY
120 '
130 '  COPYRIGHT (C) 1984
140 '    SPEECH SYSTEMS
150 '
160 ' THIS PROGRAM WILL ASK THE
170 ' USER  THE CURRENT TIME AND
180 ' DATE AND START THE CLOCK.
190 ' WHEN INSTRUCTED BY THE USER.
200 '
210 LET RTC=&HFF6E 'BASE ADDRESS OF RTC
220 DIM D(9)
230 '
240 ' REQUEST TIME AND DATE
250 '
260 CLS
270 PRINT "         SPEECH SYSTEMS"
280 PRINT "     CLOCK SETTIME PROGRAM"
290 PRINT@136,"MONTH (1-12)";:INPUT D(8)
300 IF D(8) < 1 OR D(8) > 12 THEN 290
310 PRINT@168,"DAY OF MONTH (1-31)";:INPUT D(7)
320 IF D(7) < 0 OR D(7) > 31 THEN 310
330 PRINT@200,"YEAR (0-99)";:INPUT D(9)
340 IF D(9) < 0 OR D(9) > 99 THEN 330
350 PRINT @232, "DAY OF WEEK (SUN=1)";:INPUT D(6)
360 IF D(6) < 1 OR D(6) > 7 THEN 350
370 PRINT@264,"HOUR (0-23)";:INPUT D(4)
380 IF D(4) < 0 OR D(4) > 23 THEN 370
390 PRINT@296,"MINUTE (1-59)";:INPUT D(2)
400 IF D(2) < 0 OR D(2) > 59 THEN 390
410 PRINT@328,"SECOND (1-59)";:INPUT D(0)
420 IF D(0) < 0 OR D(0) > 59 THEN 410
430 PRINT @384,"IS THIS CORRECT (Y/N)";
440 INPUT A$
450 IF A$ = "N" THEN 260
460 IF A$<>"Y" THEN 430
470 '
480 ' STORE DATE AND TIME
490 '
500 PRINT "HIT ANY KEY TO START"
510 IF INKEY$ = "" THEN 510
520 POKE RTC, 11    'CALL REG B
530 POKE RTC +1,&H80'STOP CLOCK
540 FOR I = 0 TO 9
550 POKE RTC, I
560 POKE RTC + 1, D(I)
570 NEXT I
580 POKE RTC, 10    'CALL REG A
590 POKE RTC +1,&H20'XTAL TYPE
600 POKE RTC, 11    'CALL REG B
610 POKE RTC +1,&H07'24 HOUR CLOCK
620 END

Getting Started

CLOCK
This program displays on the screen the time and date.


When the program starts, it first looks to see if the TRANSLATOR program (32K version) is loaded. The
TRANSLATOR is a text to speech program that we supply with our speech synthesizers, the VOICE and
SUPER VOICE. If this program is loaded, the time will be spoken each and every hour of the day. In
addition, hitting any key on the keyboard will force the speech synthesizer to speak the time. You may
wish to have the speech synthesizer speak the time more or less frequently (i.e. every 15 minutes).

We trust that an examination of the program and a little ingenuity will a1low you to perform this customizing.

If you do not have our speech synthesizer, the date and time will merely be displayed.

100 '           CLOCK
110 '       BY RICH PARRY
120 '
130 '     COPYRIGHT (C) 1984
140 '       SPEECH SYSTEMS
150 '
160 'A PROGRAM TO READ THE DATE/
170 'TIME OF THE RTC AND DISPLAY
180 'IT ON THE SCREEN.
190 CLS
200 '
210 CLEAR100,&H5EFF
220 ADR = &H5F00
230 BUF = ADR + 28
240 DEFUSR0 = ADR
250 IF PEEK(&H6000)=127 AND PEEK(&H6005)=32 THEN TALK=1 ELSE TALK=0
260 IF TALK=1 THEN DEFUSR1=&H6005 : GOTO 330
270 PRINT"THE TRNSLATE PROGRAM WHICH"
280 PRINT"COMES WITH OUR SPEECH"
290 PRINT"SYNTHESIZER IS NOT LOADED."
300 PRINT"THEREFORE, THIS PROGRAM"
310 PRINT"WILL NOT SPEAK THE TIME"
320 GOTO 350
330 PRINT"TRNSLATE IS LOADED, THIS"
340 PRINT"PROGRAM WILL SPEAK THE TIME"
350 PRINT:PRINT:PRINT
360 PRINT"HIT ANY KEY TO START"
370 IF INKEY$="" THEN 370
380 '
390 ' POKE ML PROGRAM IN DATA
400 ' DATA STATEMENTS IN MEMORY
410 '
420 FOR I=0 TO 27
430 READ DT
440 POKE ADR + I, DT
450 NEXT I
460 CLS
470 TREF=9999  'ANY NON REAL REFERENCE VALUE
480 SEC = 9999 ' ANY NON REAL SECOND VALUE
490 A=USR(0)
500 IF INKEY$<>"" THEN TREF=9999 : SEC=9999
510 IF PEEK(BUF+9) = SEC THEN 490
520 SEC = PEEK(BUF+9)  ' SAVE FOR FUTURE COMPARISON
530 YR = PEEK(BUF + 0)
540 MO = PEEK(BUF + 1)
550 DM = PEEK(BUF + 2)
560 DW = PEEK(BUF + 3)
570 HR = PEEK(BUF + 5)

Precision Time Module

580 MI = PEEK(BUF + 7)
590 SE = PEEK(BUF + 9)
600 PRINT @300,"";
610 PRINT USING "##:##:##"; HR; MI; SE;
620 PRINT @196,"";
630 ON DW GOSUB 860,800,810,820,830,840,850
640 ON MO GOSUB 870,880,890,900,910,920,930,940,950,960,970,980
650 PRINT STR$(DM);
660 PRINT ", ";
670 PRINT YR;
680 IF TALK=0 OR TREF=HR THEN 750
690 TREF = HR : ' UPDATE REFERENCE
700 X$=USR1("THE TIME IS")
710 X$=USR1(STR$(HR))
720 X$=USR1("HOURS")
730 X$=USR1(STR$(MI))
740 X$=USR1("MINUTES")
750 GOTO 490
760 DATA &H30,&H8D,&H00,&H18,&HC6,&H09,&H86,&H0A
770 DATA &HB7,&HFF,&H6E,&HB6,&HFF,&H6F,&H2B,&HF6
780 DATA &HF7,&HFF,&H6E,&HB6,&HFF,&H6F,&HA7,&H80
790 DATA &H5A,&H2C,&HEB,&H39
800 PRINT "MONDAY   "; : RETURN
810 PRINT "TUESDAY  "; : RETURN
820 PRINT "WEDNESDAY"; : RETURN
830 PRINT "THURSDAY "; : RETURN
840 PRINT "FRIDAY   "; : RETURN
850 PRINT "SATURDAY "; : RETURN
860 PRINT "SUNDAY   "; : RETURN
870 PRINT "JANUARY  "; : RETURN
880 PRINT "FEBRUARY "; : RETURN
890 PRINT "MARCH    "; : RETURN
900 PRINT "APRIL    "; : RETURN
910 PRINT "MAY      "; : RETURN
920 PRINT "JUNE     "; : RETURN
930 PRINT "JULY     "; : RETURN
940 PRINT "AUGUST   "; : RETURN
950 PRINT "SEPTEMBER"; : RETURN
960 PRINT "OCTOBER  "; : RETURN
970 PRINT "NOVEMBER "; : RETURN
980 PRINT "DECEMBER "; : RETURN

Getting Started

NUCLEUS
This is not meant to be a complete BASIC program but rather it is the nucleus or starting point of a typical BASIC program
that makes use of time and date. Any programs written in BASIC that need to access the time and date resources will
need to have this code included since the clock machine code must be placed somewhere in memory.

Since the machine language program is completely position independent, it can handily be placed anywhere in memory. In
the example shown, we have placed it at the top of memory for a 32K/64K Color Computer system. This is probably the
best place for for it in most situations and as a result it should not be necessary to alter the program NUCLEUS very often.

Let’s look at the program in detail to see just how the PRECISION TIME MODULE works.

Precision Time Module

LINE 220

220 CLEAR100,&H7EFF

In this line we are reserving memory for the machine language program that we are about to POKE into memory.

In the example, we are actually reserving 256 bytes when really only 38 bytes are necessary (call us wasteful if you like).

LINE 230 - 240

230 ADR = &H7F00
240 BUF = ADR + 28

In line 230 we are specifying the beginning of the machine language program and in line 240 we are specifying the
beginning of the I0 byte buffer area where the time and date are placed by the machine language program.


The buffer area is 28 bytes after the starting location since the machine language program requires 28
bytes starting at location zero. An examination of the memory map diagram shown on a separate page
should make this a bit clearer. The point to remember is that the machine language program requires
28 bytes.

Each time the machine language code is called it places the time and date at the very end of itself, a place we call the
buffer area.

LINE 250

250 DEFUSR0 = ADR

Line 250 tells BASIC where we have placed the machine language subroutine.

LINES 290 - 360

290 FOR I = 0 TO 27
300 READ DT
310 POKE ADR + I, DT
320 NEXT I
330 DATA &H30,&H8D,&H00,&H18,&HC6,&H09,&H86,&H0A
340 DATA &HB7,&HFF,&H6E,&HB6,&HFF,&H6F,&H2B,&HF6
350 DATA &HF7,&HFF,&H6E,&HB6,&HFF,&H6F,&HA7,&H80
360 DATA &H5A,&H2C,&HEB,&H39

Lines 290 to 320 merely POKE’s into memory the machine language subroutines which are contained in the BASIC DATA
statements on lines 330 to 360.

LINE 400

400 A=USR(0)

We are now really done, all we need do to find the current time and date is to call the subroutine (line 400) and then PEEK
into the buffer memory for the time and date values (see lines 440 through 500).

We believe that a little study of the BASIC program NUCLEUS and the memory map diagram entitled "Format For Date &
Time in Memory" should give you all you need to take advantage of the PRECISION TIME MODULE.

Getting Started

100 '
110 '                 NUCLEUS
120 '
130 ' THIS PROGRAM CONTAINS THE NEUCLEUS OF A BASIC
140 ' PROGRAM THAT USES THE PRECISION TIME MODULE
150 ' IT IS INTENDED AS A STARTING POINT FOR
160 ' WRITING YOUR OWN BASIC PROGRAM THAT REQUIRES
170 ' THE TIME AND DATE
180 '
190 '
200 ' SAVE SPACE FOR ML PROGRAM
210 '
220 CLEAR100,&H7EFF
230 ADR = &H7F00
240 BUF = ADR + 28
250 DEFUSR0 = ADR
260 '
270 ' POKE ML PROGRAM IN MEMORY
280 '
290 FOR I = 0 TO 27
300 READ DT
310 POKE ADR + I, DT
320 NEXT I
330 DATA &H30,&H8D,&H00,&H18,&HC6,&H09,&H86,&H0A
340 DATA &HB7,&HFF,&H6E,&HB6,&HFF,&H6F,&H2B,&HF6
350 DATA &HF7,&HFF,&H6E,&HB6,&HFF,&H6F,&HA7,&H80
360 DATA &H5A,&H2C,&HEB,&H39
370 '
380 ' CALL FOR DATE AND TIME
390 '
400 A=USR(0)
410 '
420 ' PEEK INTO MEMORY FOR TIME
430 '
440 YR = PEEK(BUF + 0) :' YEAR
450 MO = PEEK(BUF + 1) : 'MONTH
460 DM = PEEK(BUF + 2) : ' DAY OF THE MONTH
470 DW = PEEK(BUF + 3) : ' DAY OF THE WEEK
480 HR = PEEK(BUF + 5) : ' HOUR
490 MI = PEEK(BUF + 7) : ' MINUTES
500 SE = PEEK(BUF + 9) : ' SECOND

Precision Time Module

The Machine Language Program
One may ask why it is program that we must use a machine language to find out the time and date. The answer lies in the
fact that BASIC is too slow. There is an instant in time lasting a few milliseconds in which the PRECISION TIME MODULE is
rolling over. Rolling over means that once a second all the internal registers of the PRECISION TIME MODULE ic
(MC146818 ) are changing. Although it is fast, it is not instantaneous. This means that there is a period in time when the
data is invalid.

We must therefore examine the IC to find out if the data is valid. We do this by examining the UPDATE IN PROGRESS bit of
one of the registers. An examination of the program will show that we loop until the data is valid and when it is ready we
then read the time and date and store it in memory in the buffer area at the end of the program. The process of reading
the time and date and storing must be done very rapidly, therefore a machine language program is used.

        NAM     NUCLEUS
*
* A POSITION INDEPENDENT PROGRAM TO PROCESS THE DATE AND TIME OF
* THE MC146818 AND STORE IT IN A BUFFER WHICH CAN BE
* USED IN A BASIC OR MACHINE LANGUAGE PROGRAM
*
* THE DALLAS DS1287 IS COMPATIBLE WITH THE MOTOROLA MC146818 RTC. IT
* CONTAINS AN INTEGRATED CRYSTAL CLOCK SOURCE AND LITHIUM BATTERY
*
* ADDRESS MAP & REGISTER DATA
* ===========================
*
*  RTC +ADDR    FUNCTION        HEX     BUFFER #
*       0       SECONDS         00      9
*       1       SECONDS ALARM   01      8       (NOT USED)
*       2       MINUTES         02      7
*       3       MINUTE ALARM    03      6       (NOT USED)
*       4       HOURS           04      5
*       5       HOUR ALARM      05      4       (NOT USED)
*       6       WEEK DAY        06      3
*       7       MONTH DAY       07      2
*       8       MONTH           08      1
*       9       YEAR            09      0
*      10       REGISTER A      0A      N/A
*      11       REGISTER B      0B      N/A
*      12       REGISTER C      0C      N/A
*      13       REGISTER D      0D      N/A
*
* REGISTER A FUNCTIONS
*
*  B7   B6   B5   B4   B3   B2   B1   B0
*  UIP  DV2  DV1  DV0  RS3  RS2  RS1  RS0
*
* BIT 7 - UIP UPDATE IN PROGRESS. READ ONLY BIT. LOGIC "1" = UPDATE IN
*         PROGRESS. IT MUST BE AT LOGIC "O" TO STORE A VALUE IN THE RTC
*
* BITS 6,5,4 - TIME BASE DIVIDER (ALLOWED VALUES 0,1,2)
*              1 = 4.194304 MHZ
*              2 = 1.048576 MHZ
*              3 = 32.768 KHZ
*              THESE BITS ARE NOT REQUIRED BY THE DS1287
*
* BITS 3,2,1,0 - RATE SELECTOR, USED TO SELECT OUTPUT SQUARE WAVE
*              HEX 00 = OUTPUTS DISABLED
*              HEX 01 = PIR 30.517uS SQW 32.768 KHZ
*              ..  ..
*              HEX FF = PIR 500MS SQW 2 HZ
*              THESE BITS ARE NOT REQUIRED BY THE DS1287
*

*
* REGISTER B FUNCTIONS
*
*  B7   B6   B5   B4   B3   B2   B1   B0
*  SET  PIE  AIE  UIE  SQWE DM  24/12 DSE
*
* SET  1 = ABORT THE UPDATE CYCLE AND ALLOW A WRITE TO OCCUR
*      0 = NORMAL OPERATION
*
* PIE  1 = PERIODIC INTERRUPT. IT DRIVES THE IRQ PIN LOW FOR A PERIOD DEFINED
*          BY REGISTER A BITS 3,2,1,0.
*      0 = BLOCKS THE PIE FROM CHANGING THE IRQ STATE.
*
* AIE  1 = ALLOWS AN ALARM SETTING TO GENERATE AN IRQ
*      0 = DISALLOWS AN ALARM SETTING TO GENERATE AN IRQ
*
* UIE  1 = UPDATE ENDED INTERRUPT ALLOWS THE UPDATE ENDED FLAG TO GENERATE
*          AN IRQ
*      0 = PREVENTS UIE FLAG FROM GENERATING AN IRQ
*
* SQWE 1 = ALLOWS THE SQUARE WAVE RATE SELECT TO TOGGLE THE SQW PIN AT THE
*          RATE SELECTED BY REGISTER A.
*      0 = HOLDS THE SQW PIN LOW.
*
* DM   1 = BINARY DATA MODE
*      0 = BCD DATA MODE
*
* 24/12 1 = 24 HOUR MODE
*       0 = 12 HOUR MODE
*
* DSE  1 = ENABLE DAYLIGHT SAVINGS MODE
*      0 = DISABLE DAYLIGHT SAVINGS MODE
*
*
* REGISTER C FUNCTIONS
*
*  B7   B6   B5   B4   B3   B2   B1   B0
*  IRQF PF   AF   UF   0    0    0    0
*
* REGISTER C IS READ ONLY AND CONTAINS INTERRUPT FLAG INFORMATION
*
* IRQF - IRQ FLAG ANYTIME THE IRQF FLAG IS "1", THE IRQ PIN IS DRIVEN LOW.
* PF   - PERIODIC INTERRUPT FLAG
* AF   - ALARM FLAG
* UF   - UPDATE FLAG
*
* REGISTER D ONLY CONTAINS 1 ACTIVE BIT. THE MSB OF THE D REGISTER
* CONTAINS THE "VRT" VALUE INDICATING THAT THE RTC HAS GOOD POWER
* AND THE THE IT IS NO LONGER IN POWER DOWN MODE. A "0" IS READ WHEN
* THE POWER SENSE PIN IS LOW.
*
*
       ORG     $7000
RTCAD  EQU     $FF6E   * RTC MUX ADDRESS
RTCDT  EQU     RTCAD+1 * RTC DATA ADDRESS
*
*
START  LEAX    BUFFER,PCR     * PREPARE THE BUFFER
       LDB     #9             * START LOADING THE BUFFER AT LOCATION 9
                              * DECREMENT TILL WE REACH 0
                              * IT WILL LOAD SECONDS (9) THOUGH YEAR(0)
CKUIP  LDA     #10            * PREPARE TO SELECT REGISTER A
       STA     RTCAD          * STORE REGISTER SELECTION IN THE RTC

       LDA     RTCDT          * GET REGISTER CONTENTS
       BMI     CKUIP          * IS THE UPDATE BIT SET? YES, LOOP UNTIL NOT
       STB     RTCAD          * NO, THEN UPDATE VALUE
       LDA     RTCDT          * READ SELECTED DATA FROM RTC
       STA     0,X+           * STORE IT IN THE LOCAL BUFFER
       DECB                   * MOVE TO NEXT REGISTER AND BUFFER LOCATION
       BGE     CKUIP          * IF THE BUFFER VALUE IS NOT 0, LOOP
       RTS                    * BUFFER VALUE = 0, WE ARE DONE
*
BUFFER RMB     10             * RESERVE 10 LOCATIONS FOR THE BUFFER
       END     START

Battery Backup
The PRECISION TIME MODULE contains a nickel cadmium battery to keep the clock running during power those times that
the computer is off. The battery should last approximately 1 month without a charge. The battery is rechargeable which
means you should not have to replace it. Should the battery need recharging, merely place it in the computer for at least 8
hours.

Where is the PRECISION TIME MODULE in Memory
The PRECISION TIME MODULE is a memory mapped hardware peripheral and as a result it must be placed somewhere in
memory. It requires only 2 bytes of address space. We have decided to place the PRECISION TIME MODULE at $FF6E and
$FF6F. We know of no other peripheral for the Color Computer located here and as a result the PRECISION TIME MODULE
should not interfere with any other hardware device.

Powering Down the PRECISION TIME MODULE
You may notice sometimes that an un-powered PRECISION TIME MODULE will show lost time or perhaps even gives an
invalid time when powered up again.

This undesired result does have a root cause. When shutting the Color Computer off and power is removed from MC6809
there is a brief moment where it can do almost anything… including writing to the memory addresses used by the
PRECISION TIME MODULE. To help avoid this issue, the PRECISION TIME MODULE does have a circuit that helps detect
when the computer is about to lose power but it does have its limitations and the results cannot be guaranteed to work
every single time. There will be times when the time is incorrect the next time the computer is turned on.

Advanced Information for Advanced Users
The following pages contain information that will be of use to users with a good background in machine language
programming and a modicum of knowledge in electronics. We at SPEECH SYSTEMS have developed a powerful PRECISION
TIME MODULE that is capable of many applications. To describe all the features of the PRECISION TIME MODULE would
require a much larger manual. We hope the unit is easy to use for everyone, and for those advanced users, we supply the
basic technical specifications of the chip and a schematic in the hopes that they will supply the additional effort to
customize the PRECISION TIME MODULE to their particular application.

Interrupts
The PRECISION TIME MODULE IC (MC146818 ) is capable of supplying interrupts to the computer (6809). When the
PRECISION TIME MODULE leaves the factory, the jumper which connects the interrupt line, PIN 19 of the chip is left
vacant so there is no connect to the /NMI line of 6809, PIN 4. Should you wish to connect the PRECISION TIME MODULE for
an interrupt based application, you must open the PRECISION TIME MODULE case and solder a jumper wire connecting
the two holes provided. The two plated through holes are located slightly down and to the right of pin 1 on the edge
connector. Please note that only the advanced programmer should consider making use of this feature.

 Proceed with caution. Opening the PRECISION TIME MODULE will void the warranty. You are expected
to know what you are doing.

Adjusting The PRECISION TIME MODULE
The PRECISION TIME MODULE is calibrated at the factory. However, it may become necessary to fine tune the unit.
Looking at the PRECISION TIME MODULE so the connector end that plugs into the color computer faces left, you will find
an adjustable "trimmer capacitor" in the extreme upper left hand corner. Using a small screwdriver turn the capacitor dial
slightly in one direction and record which way you turned it and how much. You will then need to wait a day or more to see
if you turned it the correct amount and in the correct direction to get better crystal timing performance.


The adjustment dial "wraps around", therefore after turning it in one direction for more that 360
degrees you will be back to where you started. It also should be understood that opening the unit to
make this adjustment voids the warranty.

Schematic Diagram

Hardware Details
The MC146818 provided clock and calendar functions with 50 bytes of user RAM memory, and was designed to work with
either the Motorola or Intel processor timing. These features could be made nonvolatile by providing a backup battery on
the board to maintain the VDD supply. For timekeeping,the MC146818 required a number of discrete components on the
PCB along with a crystal to provide an input frequency. The input frequencies could be 32.768 kHz, 1.048576 MHz, or
4.194304 MHz.

Time Module Register Map

Buffer Allocation

RTC +ADDR FUNCTION HEX BUFFER #

0 SECONDS 00 9

1 SECONDS ALARM 01 8 (NOT USED)

2 MINUTES 02 7

3 MINUTE ALARM 03 6 (NOT USED)

4 HOURS 04 5

5 HOUR ALARM 05 4 (NOT USED)

6 WEEK DAY 06 3

7 MONTH DAY 07 2

8 MONTH 08 1

9 YEAR 09 0

10 REGISTER A 0A N/A

11 REGISTER B 0B N/A

12 REGISTER C 0C N/A

13 REGISTER D 0D N/A

FORMAT FOR DATE/TIME IN MEMORY

OS9 Support
Included on the following pages is a source listing that can be used to interface the Precision Time Module real time clock
with the OS9 Operating System. SPEECH SYSTEMS supplies this listing on an AS-IS basis. SPEECH SYSTEMS assumes that
you are familiar with OS9 and the related assembler programs to make use of this OS9 Specific code.

